
PROBLEMAS ANALISADOS NO ANSYS 5.0

Estrutura Reticulada

Constantes do problema :

� M�odulo de elasticidade:

E = 21� 105 Kgf=cm2 = 21� 109 Kgf=m2

� �Area da sec�c~ao transversal:

A = 1 cm2 = 10�4m2

!PRE-PROCESSADOR

/PREP7

/TITLE,ESTRUTURA RETICULADA

ET,1,LINK1 **Tipo do elemento

EX,1,21E9 **Modulo de elasticidade

R,1,1E-4 **Area

N,1 **Definicao dos nos

N,2,3

N,3,3,8

N,6,12,8

NPLOT

FILL,3,6

NPLOT

NGEN,2,3,5,5,1,0,-2

NPLOT

NGEN,2,3,4,4,1,0,-4

NPLOT

ELIS

E,1,3 **Definicao dos elementos

E,2,3

E,2,7

E,3,7

E,3,4

EPLOT

EGEN,3,1,5,5,1

EPLOT

E,4,7

E,4,8

E,5,8

E,8,7

E,6,8

EPLOT

D,1,UX,0,0,2,1,UY **Condicoes de contorno para os nos 1 e 2

DLIS

F,3,FX,1.5 **Forca em x para o no 3

F,3,FY,-1,0,6,1 **Forca em y para os nos 3 a 6

FLIS

/PBC,ALL,1

EPLOT

SAVE **Salva dados

FINI **Abandona PREP7

!SOLVER

/SOLU

SOLVE **Resolve modelo

FINI **Abandona SOLVER

!POS-PROCESSADOR

/POST1

ETABLE,FB,SMISC,1 **Forca nas barras no sistema do elemento

ETABLE,TEN,LS,1 **Tensao nas barras

SET **Le resultados

PLDI,1 **Plot da geometria deformada

PRDIS **Deslocamentos nodais

PRETAB,FB,TEN **Forcas para os nos e tensao

PRESOL,F **Forcas decompostas para os nos

PLETAB,TEN **Apresenta as tensoes nos elementos

PLETAB,FB **Apresenta forca ao longo dos elementos

FINI **Abandona POST1

/EXIT **Abandona ANSYS

Deforma�c~ao em Vigas

Constantes do problema :

� M�odulo de elasticidade:

E = 21� 106 tf=m2

� �Area da sec�c~ao transversal:

A = 3; 75� 10�2m2

� Momento de in�ercia em rela�c~ao ao eixo z:

Jz =
bh3

12
= 15�25

3

12
= 1; 953� 104m4

� Altura do per�l:

h = 25 cm

X

!PRE-PROCESSADOR

/PREP7

/TITLE,VIGA - 10 ELEMENTOS

EX,1,21E6 **Modulo de elasticidade do material da viga

ET,1,BEAM3 **Tipo do elemento: viga

R,1,3.75E-2,1.953E-04,25E-2 **Area, Izz, h

N,1,0,0 **Definicao dos nos

N,11,2,0

NPLOT

FILL **Gera nos intermediarios

/PNUM,NODE,1

NPLOT

NLIS

E,1,2 **Definicao dos elementos

EGEN,10,1,1,1,1 **Gera o restante dos elementos

/PNUM,ELEM,1

EPLOT

D,1,ALL,0 **Engastado no no' 1

F,11,FY,-3 **Forca de -3t no no' 11

F,11,MZ,-2 **Momento de -2 t.m no no' 11

EP,1,1,1,0,10,1 **Carga distribuida de -1 t/m na face do elemento 1

/PBC,ALL,1 **Mostra condicoes de contorno

/VIEW,1,1,1,1 **Vista em perspectiva

EPLO **Plot dos elementos

SAVE **Salva dados

FINI **Abandona prep7

!SOLVER

/SOLU

SOLVE

FINI **Abandona SOLVER

/COM,POS-PROCESSADOR

/POST1

ETABLE,FXI,SMISC,1 **Forca de reacao nodal na direcao x - no i

ETABLE,FYI,SMISC,2 **Forca de reacao nodal na direcao y - no i

ETABLE,MZI,SMISC,6 **Momento fletor mz nos pontos nodais - no i

ETABLE,FYJ,SMISC,8 **Le forca de reacao nodal na direcao y - no j

ETABLE,MZJ,SMISC,12 **Le momento fletor mz nos pontos nodais - no j

ETABLE,SDI,LS,1 **Tensao devido ao esforco normal - nos i e j

ETABLE,SDJ,LS,4

ETABLE,SBI,LS,2 **Tensao de flexao para os nos i e j

ETABLE,SBJ,LS,5

ETABLE,SDPI,NMISC,1 **Tensao de flexao+normal para os nos i e j

ETABLE,SDPJ,NMISC,3

ETABLE,SDMI,NMISC,2 **Tensao de flexao-normal para os nos i e j

ETABLE,SDMJ,NMISC,4

SET **Le resultados

PLDI,1 **Grafico da geometria deformada

PRDIS **Imprime deslocamentos e rotacoes

PRETAB,FXI,FYI,FYJ,MZI,MZJ **Imprime esforcos nodais

PRETAB,SDI,SDJ,SBI,SBI **Tensoes axiais e de flexao

PRETAB,SDMI,SDMJ,SDPI,SDPJ **Tensoes resultantes

PRESOL,F **Imprime forcas de reacao em cada elemento

PRESOL,M **Imprime momentos de reacao em cada elemento

PLLS,FYI,FYJ **Forcas nodais na direcao y

PLLS,MZI,MZJ **Momentos nodais em z

PLLS,SDPI,SDPJ **Tensoes resultantes

PLLS,SDMI,SDMJ

PLETAB,SDPI **Tensoes nos elementos

PLETAB,SDPJ

PLETAB,SBI

PLETAB,SBJ

FINI **Abandona POST1

/EXIT **Abandona Ansys

P�ortico

Propriedades e constantes :

� M�odulo de elasticidade das vigas:

E = 21; 0� 106 tf=m2

� �Area da sec�c~ao transversal das vigas:

A = 3; 75� 10�2 m2

� Momento de in�ercia em rela�c~ao ao eixo z:

Jz = 1; 953� 104m4

� M�odulo de elasticidade das barra:

E = 7; 0� 106 tf=m2

� �Areas da sec�c~ao transversal das barras inclinadas:

A = 2; 0 cm2

� �Areas da sec�c~ao transversal das barras horizontais:

A = 1; 0 cm2

! Estrutura com varios materiais e elementos

!

!PRE-PROCESSAMENTO

/prep7

/title,Portico

!

! Configuracao

!

/pnum,node,1

/pnu,elem,1

!

! Tipos de Elementos/Constantes/Materiais

!

et,1,link1 **elemento de barra

et,2,beam3 **elemento de viga plana

r,1,3.75e-2,1.953e-4,25e-2 **area, Izz, h

r,2,1e-4 **area

r,3,2e-4 **area

ex,1,7e6 **modulo de elasticidade do aluminio

ex,2,21e6 **modulo de elasticidade do aco

nuxy,1,0.28 **coef. poisson aluminio=0.28

nuxy,2,0.3 **coef. poisson aco=0.3

!

! Nos

!

n,1

n,2,.20

n,3,0,.20

n,4,.20,.20

nplot

!

! Elementos de aluminio - barras

!

type,1 **inicializa os conjuntos de constantes a serem utilizados

real,2

mat,1

e,1,2

real,3

e,1,4

real,2

e,3,4

eplot

!

! Elementos de aco - vigas

!

type,2 **inicializa os conjuntos de constantes a serem utilizados

real,1

mat,2

e,1,3

e,4,2

eplot

!

! Geracao do restante dos nos e elementos

!

ngen,5,2,3,4,1,0,0.20 ** Gera os nos a partir de 3 e 4

egen,5,2,4,5,1 ** Copia elementos de aco

egen,5,2,2,2,1 ** Copia elementos de aluminio

egen,5,2,3,3,1 ** Copia elementos de aluminio

eplot

!

! Colocacao das restricoes

!

d,1,all,0,0,2,1 ** Engastado na base

!

! Colocacao das forcas

!

f,11,fx,2,0,12,1 ** Forca em X nos nos 11 e 12 de 2t

f,11,fy,5 ** Forca em Y nos nos 11 e 12 de 5t

f,12,fy,-3 ** Forca em Y nos nos 11 e 12 de -3t

!

!Graficos

!

/view,1,1,1,1 **Vista em perspectiva

/pbc,all,1

/pnum,type,1 **Diferencia tipos de elementos

eplot

/pnum,real,1 **Diferencia constantes reais

eplot

/pnum,mat,1 **Diferencia materiais

eplot

!

! Grava arquivos e sai do PREP7

!

save

fini

!SOLVER

/solu

solve

fini

!POS-PROCESSAMENTO

/post1

etable,imz,smisc,6 **Momentos fletores para os nos i e j

etable,jmz,smisc,12

etable,ibnd,ls,2 **Tensao de flexao para os nos i e j

etable,jbnd,ls,5

etable,ten,ls,1 **Tensao nos elementos de barra

set **Le resultados

pldisp,1 **Geometria deformada

plls,imz,jmz **Momentos fletores para os nos i e j

plls,ibnd,jbnd **Tensao de flexao nas vigas

pletab,ten **Tensao nas barras

fini **Abandona POST1

/exit **Abandona Ansys

Estudo de um Eixo

Propriedades e Constantes :

� M�odulo de Elasticidade:

E = 21� 103 MPa

� �Area da sec�c~ao transversal:

A = �d2

4

� Momentos de in�ercia em rela�c~ao aos eixos x e y:

Jy = Jz =
�d4

64

� For�ca axial:

Fa = 2; 4 KN

� For�ca radial:

Fr = 3; 3 KN

� For�ca tangencial:

Ft = 8; 0 KN

� Torque: T = 911; 2Nm;

� Diâmetro da engrenagem:

d = 227; 8mm

� Diâmetro do trecho AB:

d1 = 50mm

� Diâmetro do trecho BC:

d2 = 45mm

!PRE-PROCESSADOR

/PREP7

/TITLE,Eixo com engrenagem

!

! Constantes do problema

!

d1=50e-3 * [m] diametro eixo 1

d2=45e-3 * [m] diametro eixo 2

d=227.8e-3 * [m] diametro primitivo eng.

l1=150e-3 * [m] comprimento eixo1

l2=140e-3 * [m] comprimento eixo2 - 1a. parte

l3=70e-3 * [m] comprimento eixo2 - 2a. parte

l4=40e-3 * [m] largura do rebaixo do mancal B

Ft=8.0e3 * [N] Forca Tangencial

Fr=3.3e3 * [N] Forca Radial

Fa=2.4e3 * [N] Forca axial

T=911.2 * [N.m] Torque transmitido

pi=3.1415926

!

! Constante do material

!

EX,1,21e9

!

! Seleciona elemento

!

ET,1,BEAM4

!

! Define as constantes reais (geom. propert.)

!area, momentos de inercia em Z e Y, espessuras em Z e Y

R,1,pi*d1**2/4,pi*d1**4/64,pi*d1**4/64,d1,d1

R,2,pi*d2**2/4,pi*d2**4/64,pi*d2**4/64,d2,d2

!

! Geracao dos nos

!

N,1

N,16,l1

FILL

/PNUM,NODE,1

NPLOT

N,30,l1+l2

FILL

NPLOT

N,35,l1+l2+l3-l4/2

FILL

NPLOT

N,37,l1+l2+l3

FILL

NPLOT

N,39,l1+l2+l3+l4/2

FILL

NPLOT

!

! Geracao dos elementos

!

! Constantes reais 1

REAL,1 * seleciona as propriedades 1

E,1,2 * gera um elemento

EGEN,34,1,1 * gera os demais

/PNUM,ELEM,1

EPLOT

! Constantes reais 2

REAL,2 * seleciona as propriedades 2

E,35,36 * gera um elemento

EGEN,4,1,35 * gera os demais

EPLOT

!

! Condicoes de contorno

!

D,16,UX,0,0,0,0,UY,UZ

D,37,UY,0,0,0,0,UZ,ROTX,ROTY,ROTZ

!

! Carregamento

!

! Na extremidade (acoplamento)

!

F,1,MX,T * Torque no acoplamento

!

! Na engrenagem

!

F,30,FX,-Fa **Forca axial do engrenamento

F,30,FY,Fr **Forca radial (cortante)

F,30,FZ,Ft **Forca tangencial (cortante)

F,30,MX,-Ft*d/2 **Momento da Forca tangencial

F,30,MZ,Fa*d/2 **Momento da Forca axial

/PBC,ALL,1 **Mostra condicoes de contorno

/PNUM,DEFA

/VIEW,1,1,1,1

EPLO **Grafico dos elementos

SAVE **Salva dados

FINI **Abandona prep7

!SOLVER

/SOLU

SOLVE

FINI **Abandona SOLVER

!POS-PROCESSADOR

/POST1

ETABLE,SG1I,NMISC,1 **Tensoes maximas de tracao para os nos i e j

ETABLE,SG1J,NMISC,3

ETABLE,SG3I,NMISC,2 **Tensoes maximas de compressao para os nos i e j

ETABLE,SG3J,NMISC,4

ETABLE,MXI,SMISC,4 **Momento torcor para os nos i e j

ETABLE,MXJ,SMISC,10

SET **Le resultados

PLDI,1 **Geometria deformada

PLLS,SG1I,SG1J **Tensoes maximas de tracao e compresao

PLLS,SG3I,SG3J

PLETAB,SG1I **Tensao de tracao no elemento

PLLS,MXI,MXJ **Momento torcor

PRETAB,MXI,MXJ **Imprime momento torcor para os nos i e j

PRDIS **Imprime deslocamentos e rotacoes

PLNS,U,X **Deslocamentos em X

FINI **Abandona post1

/EXIT **Abandona Ansys

Viga - Problema de Estado Plano de Tens~ao

Constantes do Problema :

� M�odulo de elasticidade: E = 21� 105 Kgf=cm2;

� Coe�ciente de Poisson: � = 0; 3;

� Espessura: t = 1 cm.

10Kgf

10Kgf

1m

20cm

!PRE-PROCESSADOR

/PREP7

/TITLE,Viga - Elementos planos (plane42)

!

! Material

!

EX,1,21e5 * E = 21 x 10^5 kgf/cm^2

NUXY,1,0.3 * Coeficiente de Poisson

!

! Tipo Elemento

!

ET,1,PLANE42

!

! Nos

!

N,1

N,2,0,10

N,3,0,20

/PNUM,NODE,1 *Numera nos

NPLOT

!

!Gera nos adicionais

NGEN,21,3,1,3,1,25

NPLOT

!

! Elementos

!

E,1,4,5,2 * Gera o 1o. elemento

E,2,5,6,3 * Gera o 2o. elemento

EPLOT

EGEN,20,3,1,2 * Gera os demais elementos

/PNUM,ELEM,1 * Numera elementos

EPLOT

!

! Restricoes

!

D,1,ALL,0,0,3,1 * Engastado em uma extremidade

!

! Carregamento

!

F,61,FX,10 **Forca em x para o no 13

F,63,FX,-10 **Forca em x para o no 15

!Grafico da malha

/PBC,ALL,1 **Apresenta condicoes de contorno

EPLOT

SAVE **Salva modelo

FINI **Abandona prep7

!SOLVER

/SOLU

SOLVE

FINI **Abandona SOLVER

!POS-PROCESSADOR

/POST1

SET **Le arquivo de resultados file12.dat

PLDI,1 **Plot da geometria deformada

PRDIS **Imprime deslocamentos nodais

PLNSOL,S,EQV **Tensao equivalente de von Mises

PLNSTR,S,X **Tensao na direcao x

PLNSTR,S,Y **Tensao na direcao y

PLNSTR,U,X **Deslocamento em x

PLNSTR,S,XY **Tensao de cisalhamento no plano XY

PRNSOL,S,COMP **Imprime componentes de tensoes

PRNSOL,S,PRIN **Imprime tensoes principais

PRESOL,S **Impruime tensoes para os nos por elemento

FINI **Abandona post1

/EXIT **Abandona Ansys

Viga - Problema de Estado Plano de Tens~ao - Gera�c~ao Au-

tom�atica

/PREP7

ET,1,PLANE42,0,0,3

R,1,1.5

MP,EX,1,21e5

MP,NUXY,1,0.3

K,1

K,2,100

K,3,100,20

K,4,,20

A,1,2,3,4

ESHAPE,2

LESIZE,4,10

LESIZE,2,10

ESIZE,5

AMESH,1

SAVE

FINISH

/SOLU

DK,1,ALL,,,1

DK,4,ALL,,,1

SBCTRAN

FK,2,FX,10

FK,3,FX,-10

SAVE

SOLVE

FINI

/POST1

SET

PLDI,1

PLNS,S,EQV

FINI

/EXIT

Problema com Simetria

Constantes do Problema :

� M�odulo de elasticidade: E = 21� 105 Kgf=cm2;

� Coe�ciente de Poisson: � = 0; 3;

� Espessura: t = 3:2 cm.

r. 2 cm

r.  8 cm

p

!PRE-PROCESSADOR

/PREP7

/title,Chapa Circular - Simetria

!

! Material

!

ex,1,21e5 * E = 21 x 10^5 kgf/cm^2

nuxy,1,0.3 * Coeficiente de Poisson

!

! Tipo Elemento

!

et,1,PLANE42,0,0,3 * Plane42 com entrada da espessura

r,1,3.2 * Espessura = 3.2 cm

!

! Sistema de Coordenadas cilindrico

!

csys,1

!

! Nos

!

/pnum,node,1

n,1,2,0

n,4,8,0

nplot

fill

nplot

ngen,7,4,1,4,1,0,15 * Gera o restante de 15 em 15 graus

nplot

!

! Elementos

!

/pnum,elem,1

e,1,2,6,5 * Primeiro elemento

eplot

egen,3,1,1 * Copia ao longo do raio

eplot

egen,6,4,1,3 * Copia ao longo da circunferencia

eplot

!

! Simetria

! Plano X

csys,0

nsel,s,loc,x,0,0.1

dsym,symm,x,0

! Plano y

nsel,s,loc,y,0

dsym,symm,y,0

nall

!

!Grafico com condicoes de contorno

/pbc,all,1

eplot

!

! Carregamento

!

csys,1

nsel,s,loc,x,2

sf,all,press,10 * Pressao nos elementos em r=2cm de 10 kgf/cm^2

nall

/psf,pres,2

eplot

save **Salva modelo

fini **Abandona prep7

!SOLVER

/solu

solve

fini **Abandona SOLVER

!POS-PROCESSADOR

/POST1

SET **Le resultados

PLDI,1 **Geometria deformada

PRDIS **Imprime deslocamentos

PLNSOL,S,EQV **Tensao equivalente de von Mises

PLNSOL,S,X **Tensao na direcao x

PLNSOL,S,Y **Tensao na direcao y

PRNSOL,S,COMP **Componentes de tensao sx,sy,sxy...

FINI **Abandona post1

/EXIT **Abandona Ansys

M�ultiplos Carregamentos

!PRE-PROCESSADOR

/PREP7

!

! Varios Casos de Carregamentos

!

! (baseado no exemplo: "Chapa Circular - Simetria (stif 42)"

!

resume * para ler o modelo do file16.dat

eplot * mostra o modelo e o carregamento

!

! Gravar o carregamento ja' aplicado

!

lswrite

!

! Apagar todo o carregamento

!

sfdel,all,press * apaga todas as pressoes

!

! Coloca outro carregamento e grava

!

csys,1

nsel,s,loc,x,2

sf,all,press,30 * pressao de 30 kgf/cm^2 rm r=2cm

nall

f,24,fy,-100 * forca de 100 kgf no no' 24

lswrite

!

save *salva modelo

finish *abandona prep7

! Solucao

!

/solu

lssolve,1,2

fini

!

! Pos-Processamento

!

!

/POST1

!Resultados do primeiro caso de carregamento

set,1

pldisp,1 *deformada

plnsol,s,eqv *tensao equivalente de von Mises

!Resultados do segundo caso de carregamento

set,2

pldisp,1 *deformada

plnsol,s,eqv *tensao equivalente de von Mises

fini *Abandona post1

/exit *Abandona Ansys

Chapa Circular - Gera�c~ao Autom�atica

/PREP7

!Tipo do elemento - plane stress com espessura

ET,1,plane42,,,3

!Espessura

R,1,3.2

!Modulo de elasticidade e coeficiente de Poisson

MP,EX,1,21e5

MP,NUXY,1,.3

!Geometria

K,1

K,2,2

K,3,8

K,4,,2

K,5,,8

/pnum,kpoi,1

kplot

LARC,2,4,1,2

LARC,3,5,1,8

L,2,3

L,4,5

A,2,3,5,4

/pnum,line,1

lplot

LESIZE,1,,,6

LESIZE,2,,,6

ESIZE,,3

ESHAPE,2

AMESH,1

/pnum,node,1

eplot

save

fini

!Solucao

/solu

!Condicoes de simetria

DL,3,1,SYMM

DL,4,1,SYMM

!Carga distribuida na linha 1

SFL,1,PRES,10

/pbc,all,,1

/psf,press,2

lplot

sbctra

eplot

!Grava caso de carregamento 1

LSWRITE,1

!Apaga pressao para a definicao do proximo carregamento

SFLDELE,1,PRES

!Segundo caso: pressao + forca concentrada

SFL,1,PRES,5

F,10,FY,-100

LSWRITE,2

!Solucao para os dois casos de carregamento

LSSOLVE,1,2

fini

!Pos-processamento

/POST1

!Le resultados do primeiro caso de carregamento

SET,1

PLDISP,1

PLNSOL,S,EQV

!Le resultados do segundo caso de carregamento

SET,2

PLDISP,1

PLNSOL,S,EQV

fini

/exit

Geometria no Formato IGES

!Ansys (IGES->Ansys)

/aux15 * modulo de conversao

iges,modelo,igs * arquivo modelo.igs

fini * termina conversao e grava no banco de dados

!Pre'-Processamento

/prep7

!Grafico nas linhas

/view,1,1,1,1 * vista em (1,1,1)

lplot * mostra o que foi convertido

!

!Propriedades do material

ex,1,21e11

dens,1,8042

!

!Tipo do elemento - viga 3D

et,1,beam4 * 3D elastic beam

!

!Constantes reais: area, momentos de inercia,alturas

r,1,0.002738,5.2450E-5,7,995E-5,0.06,0.2 * constantes reais

rmore,0,0,1,1 * constantes para cisalhamento

!

!Geracao de malha em linha

esize,0.5 * elemento max=0.5 [m]

lmesh,all * gera automatico

!

!Seleciona os nos para aplicacao das restricoes

nsel,y,0 * seleciona y=0

d,all,all * engasta todos os nos

nall * todos os nos

!

!Carregamento

!carga distribuida

!seleciona nos e elementos com coordenada y = 5.8

nsel,s,loc,y,5.8

esln,s,1

sfbeam,all,2,press,100 * pressao de 100 N/m^2

nall

eall

!peso proprio

acel,0,10,0 * gravidade em -y (10m/s^2)

!

! no'=99, utilize o comando de selecao com pick

!

! nsel,pick

!

f,99,fx,1000 * forca de 1 kN na direcao x

f,99,fy,-2000 * forca de 2 kN na direcao -y

!

!Grafico dos elementos com as condicoes de contorno

/pbc,all,1

/psf,press,2

eplot

save * salva modelo

fini * abandona prep7

!Solver

/solu * acessa solver

solve * resolve modelo

fini * abandona solver

!Pos-processador

/post1

set * le resultados

pldisp,1 * deformada

fini * abandona post1

/exit * abandona ansys

Gera�c~ao Autom�atica de Malhas em �Areas

Constantes do Problema :

� M�odulo de elasticidade: E = 21� 105 Kgf=cm2;

� Coe�ciente de Poisson: � = 0; 3;

� Espessura: t = 0:7 cm.

!PRE_PROCESSADOR

/PREP7

/TITLE,GERACAO DE MALHA - CHAVE

!

!Keypoints

!

K,1,21

K,2,21.6,2.1

K,3,15.9,3.6

K,4,15.3,1.5

K,5,10.2.,5.1

K,6,9.6,3

/PNUM,KPOI,1

KPLO

K,7,7,6

K,8,6.5,3.8

K,9,3.5,6.5

K,10,5.75,6.6

K,11,3.9,3.1

K,12,2.5,3.3

K,13,1.4,4

K,14,1,4.6

K,15,2.8,4.6

K,16,3.5,5.1

KPLO

K,17,2.8,7

K,18,1,7

K,19,1.45,7.65

K,20,2.5,8.3

K,21,3.9,8.5

K,22,5,7.85

KPLO

!

!Areas

!

A,1,2,3,4

A,4,3,5,6

A,6,5,7,8

APLOT

!

!Linhas

!

L,8,16

L,16,9

L,9,10

L,10,7

L,8,11

L,11,12

L,12,13

L,13,14

L,14,15

L,15,16

LPLO

L,9,17

L,17,18

L,18,19

L,19,20

L,20,21

L,21,22

L,22,10

LPLO

!

!Areas definidas por linhas

AL,9,11,12,13,14

AL,11,15,16,17,18,19,20

AL,13,21,22,23,24,25,26,27

/PNUM,AREA,1

APLO

!

!Tipo do elemento - triangulo para estado plano de tensao

ET,1,PLANE2,0,0,3

!

!Espessura da chave

R,1,0.7

!

!Aplica pressao numa linha

!

SFL,2,PRESS,10

!

!Aplica restricoes em keypoints

!

DK,9,ALL,0

DK,14,ALL,0

DK,15,ALL,0

DK,16,ALL,0

DK,17,ALL,0

!

!Define tamanho e forma do elemento

ESIZE,,3 *3 divisoes por linha

ESHAPE,1

!

!Geracao de elementos nas areas

AMESH,ALL

!

!Transfere condicoes de contorno do modelo solido para malha

SBCTRAN

!

!Apresenta elementos

/PBC,ALL,1

/PSF,PRESS,2

EPLO

!

!Propriedades do material

EX,1,21E5

NUXY,1,0.3

!

!Grava modelo no banco de dados e abandona prep7

SAVE

FINI

!SOLVER

/SOLU

SOLVE *resolve modelo

FINI *abandona Ansys

!POS-PROCESSAMENTO

/POST1

SET *le resultados

PLDI,1 *deformada

PLNSOL,SX *grafico de tensao na direcao x

FINI *abandona post1

/EXIT *abandona Ansys

Gera�c~ao Autom�atica de Malha em Volumes

Constantes do Problema :

� M�odulo de elasticidade: E = 21� 105 Kgf=cm2;

� Coe�ciente de Poisson: � = 0; 3;

!PRE-PROCESSAMENTO

/PREP7

/title,Viga - Geracao Automatica em Volumes)

!

! Material

!

ex,1,21e5 * E = 21 x 10^5 kgf/cm^2

nuxy,1,0.3 * Coeficiente de Poisson

!

! Tipo Elemento

!

et,1,SOLID45 * Elemento Solid45: 3D Isopara. Solid

!

! Kpoints

!

k, 1, 0, 0 * base

k, 2, 0, 1

k, 3, 10, 0

k, 4, 10, 1

k, 5, 4.5, 1

k, 6, 5.5, 1

k, 7, 4.5, 0

k, 8, 5.5, 0

k, 9, 4.5, 9 * alma

k,10, 5.5, 9

k,11, 4.5, 10 * topo

k,12, 5.5, 10

k,13, 2, 9

k,14, 8, 9

k,15, 2, 10

k,16, 8, 10

kplot

!

! Linha para o Vdrag (numero 1)

!

k,17,0,0,40 * kpoint

l,1,17 * linha (no. 1)

/view,1,1,1,1 * Observador em (1,1,1)

kplot

!

! Areas

!

a,1,2,5,7 * base

a,5,6,8,7

a,6,4,3,8

a,5,9,10,6 * alma

a,13,15,11,9 * topo

a,11,12,10,9

a,12,16,14,10

aplot

!

! Vdrag (gera volume por extrusao de areas)

!

vdrag,1,2,3,4,5,6,1 * areas 1-6 ao longo da linha 1

vdrag,7,0,0,0,0,0,1 * area 7 ao longo da linha 1

!

! Ajusta o modelo gerado

!

numm,all * executa um merge das entidades

!

! Visualizacao

!

/pnum,default * volta numeracao ao 'default'

/type,1,2 * Hidden Lines (centroidal sort)

vplot

!

! Controle/Geracao Automatica

!

esize,2 *tamanho medio do elemento igual a 2

eshape,2 *geracao com apenas cubos

vmesh,all *gera malha

eplot *grafico da malha gerada

!

! Restricoes

!

nsel,s,loc,z,0 *seleciona todos os noz com z=0

d,all,all,0 *viga engastada numa extremidade

nall

!

! Carregamento

!

! (Carga distribuida de 40 kgf/cm no topo)

! (areas: 26, 30 e 33)

asel,s,,,26,30,4 * seleciona 26 e 30

asel,a,,,33 * adciona ao grupo a area 33

sfa,all,1,press,40 * pressao nas areas selecionadas

aplot

arall * seleciona todas as areas

aplot

!

! Transporta B.C. para o mesh

!

sbctran

!

! Visualiza

!

/pbc,all,1

/psf,press,2

eplot

!

save *Salva dados

fini *Abandona prep7

!SOLVER

/solu

solve

fini **Abandona SOLVER

! Pos-Processamento

!

/post1

set **le resultados

pldisp,1 **deformada

plnsol,s,eqv **tensao equivalente de von Mises

fini **abandona Ansys

/exit **Abandona Ansys

38



Simetria em Volumes

Constantes do Problema :

� M�odulo de elasticidade: E = 21� 105 Kgf=cm2;

� Coe�ciente de Poisson: � = 0; 3;

!PRE-PROCESSADOR

/PREP7

/title,Disco - Simetria em Volumes

!

! Material

!

ex,1,21e5 * E = 21 x 10^5 kgf/cm^2

nuxy,1,0.3 * Coeficiente de Poisson

!

! Tipo Elemento

!

et,1,solid45 * Elemento solid45: 8 node Isopara. Solid

!

! Sistema Cilindrico

!

csys,1

!

! Kpoints

!

k,1,1,0

k,2,1,90

k,3,10,0

k,4,13,0

k,5,10,90

k,6,13,90

/pnum,kpoi,1

kplot

!

! Linhas para o Vdrag (numero 1 e 2)

!

k,7,1,0,2

k,8,1,0,4

l,1,7 * linha 1

l,7,8 * linha 2

/view,1,1,1,1

kplot

!

! Areas

!

a,1,3,5,2

a,3,4,6,5

/pnum,area,1

aplot

!

! Vdrag (gera volume por extrusao de areas)

!

vdrag,1,0,0,0,0,0,1 * area 1 ao longo da linha 1

/pnum,volu,1

vplot

vdrag,2,0,0,0,0,0,1,2 * area 2 - linhas 1 e 2

vplot

!

! Compatibiliza entidades na interseccoes

!

numm,all

!

! Visualizacao

!

/pnum,default * nao apresenta a numeracao das entidades

/view,1,-1,1,1 * Observador em (1,1,1)

/type,1,2 * Hidden Lines (centroidal sort)

vplot

!

! Controle/Geracao Automatica

!

esize,3 * Tamanho medio do elemento igual a 3

eshape,2 * Elemento na forma de cubo

vmesh,all * Gera nos e elementos em todas as entidades

eplot

!

! Executa a simetria em x e y (cartesianos)

!

csys,0 **Retorna ao sistema global cartesiano

vsymm,X,all **Simetrias em X e Y

vplot

vsymm,Y,all

vplot

!

! Coerencia do modelo/malha

!

numm,all

!

! Restricoes

!

csys,1 **Ativa sistema global cilindrico

nsel,s,loc,x,1 **Seleciona todos os nos com raio igual a 1

d,all,all,0 **Engastado em uma extremidade

nall

!

! Carregamento

!

nsel,s,loc,z,4 **seleciona nos com z = 4

sf,all,press,40 **pressao de 40 kgf/cm^2 em z=4

nall **reseleciona todos os nos

!

! Visualiza malha com condicoes de contorno

!

/pbc,all,1

/psf,press,2

eplot

save **Salva modelo no banco de dados

fini **Abandona prep7

!SOLVER

/solu

solve

fini **Abandona SOLVER

!POS-PROCESSADOR

/POST1

SET **Le arquivo de resultados

/VIEW,1,1,1,1

PLDI,1 **Geometria deformada

PRDIS **Imprime deslocamentos e rotacoes

PLNSOL,S,EQV **Tensao nodal equivalente de von Mises

PLESOL,S,EQV **Tensao equivalente de von Mises no elemento

PLNSOL,S,X **Tensao nodal na direcao x

PLNSOL,S,Y **Tensao na direcao y

PLNSOL,S,XY **Tensao nodal de cisalhamento

PRNSOL **Imprime tensoes nodais

FINI **Abandona prep7

/EXIT **Abandona Ansys

Estrutura modelada por Elementos de Placa

Constantes do problema :

� M�odulo de elasticidade: E = 21� 105 Kgf=cm2 = 21� 109 Kgf=m2;

� Coe�ciente de Poisson: � = 0; 3.

! Estrutura tridimensional com elementos de casca

!

! PREPROCESSAMENTO

!

/prep7

/title,Estrutura tridimensional com casca

!

! Tipos de Elementos/Constantes/Materiais

!

et,1,shell63 ** ELASTIC QUADRILATERAL SHELL

r,1,3.15 ** Espessura

ex,1,21e6 ** E aco (kfg/cm2)

nuxy,1,0.3 ** coef. poisson aco=0.3

!

! Pontos

!

! keypoints do modelo

k,1,40

k,2,20

k,3,20,100

/pnum,kpoi,1

kplot

! keypoints usados na rotacao (veja arotate logo abaixo)

k,4,0

k,5,0,100

kplot

!

! Linhas

!

l,1,2

l,2,3

lplot

!

! Areas

!

arotate,1,2,0,0,0,0,4,5,360,8

/view,1,1,1,1

aplot

!

! Seleciona as linhas do topo

!

ksel,s,loc,y,100 ** seleciona os pontos

kplot

lslk,s,1 ** seleciona as linhas correspondentes

lplot

!

! Controla a geracao para 2 divisoes por linha

!

lesize,all,0,0,2

lplot

!

! Reseleciona todas os pontos e linhas

!

ksel,all

lsel,all

!

! Geracao da malha

!

esize,30, ** Tamanho maximo do elemento=30cm

eshape,1 ** Elementos triangulares

amesh,all ** gera malha em todas as areas

numm,all ** elimina redundancias

eplot

!

! Aplica as restricoes no topo

!

nsel,s,loc,y,100 **seleciona todos os nos com y=100

nlis

nplot

d,all,all **aplica restricoes

nall **reseleciona todos nos

nplot

!

! Aplica a pressao na base

!

ksel,s,loc,y,0

lslk,s,1

asll,s,1

aplot

sfa,all,1,press,-100

/psf,press,2

aplot

!

! Reseleciona tudo novamente

!

ksel,all

lsel,all

asel,all

!

!

! Transfere solido->malha

sbctrans

!

! Visualiza modelo

!

/view,1,1,-1,1

/pbc,all,1

eplot

!

! Grava arquivos e sai do PREP7

!

save

fini

!

!SOLVER

!

/solu

solve

fini

!

! POS-PROCESSAMENTO

!

/post1

set

pldisp,1 **Deformada

plns,s,eqv **Tensao nodal equivalente de von Mises

fini **Abandona post

/exit **Abandona Ansys

An�alise Dinâmica de uma Placa

Constantes do problema :

� M�odulo de elasticidade: E = 21� 105 Kgf=cm2 = 21� 109 Kgf=m2;

� Densidade: � = 7; 85� 103 Kgf=cm2;

� Coe�ciente de Poisson: � = 0; 3.

!

! Analise dinamica de uma placa

!

! PREPROCESSAMENTO

!

/prep7

/title,Analise dinamica de uma placa

!

! Tipos de Elementos/Constantes/Materiais

!

et,1,shell63 ** ELASTIC QUADRILATERAL SHELL

r,1,3.5 ** Espessura=5cm

ex,1,21e6 ** E aco (kfg/cm2)

dens,1,7.85e-3 ** Densidade=7.85e-3 kg/cm3

nuxy,2,0.3 ** coef. poisson aco=0.3

!

! Pontos

!

k,1,0, 0

k,2,40,0

k,3,0, 40

k,4,40,40

/pnum,kpoi,1

kplot

!

! Area

!

a,1,2,4,3

/pnum,area,1

aplot

!

! Geracao da malha

!

esize,5 ** Tamanho maximo do elemento=5cm

eshape,2 **Elementos quadrangulares

amesh,all ** gera malha em todas as areas

!

! Visualiza modelo

!

/pnum,elem,1

eplot

!

! Grava arquivos e sai do PREP7

!

save

fini

!

! SOLUCAO

!

/solu

!Tipo da analise

!

antype,modal

modopt,reduc,24

total,50

!solve

fini **Abandona solver

!

!Expansao dos modos

/solu

expass,on **Passo de expansao

mxpand,12 **Numero de modos a serem expandidos

solve

! POS-PROCESSAMENTO

!

/post1

!

! Oitavo modo como exemplo

!

set,1,7 ** le valores

/window,1,left ** divide em duas janelas, 1=esquerda

/window,2,right ** janela 2=direita

/window,2,off ** desliga janela 2

/view,1,1,1,1 ** ponto do obs. em (1,1,1) na janela 1

pldisp,1 ** desenha deformada e original

/window,1,off ** desliga janela 1

/window,2,on ** liga janela 2

/noerase ** nao apaga a tela antes do plot

/view,2,1,1,1 ** ponto do obs. em (1,1,1) na janela 2

plvect,disp ** desenha os vetores do deslocamento

/erase ** volta a apagar

/fini ** abandona post1

/exit ** abandona Ansys

